
Administrivia

 Lab 1 due tonight at 11:59pm

 You have late days available

 Homework 2 due next Friday (10/20)

 Lab 2 (x86‐64) released on Tuesday (10/17)

 Due on 10/27

Review:  Operand types

 Immediate: Constant integer data
 Examples:  $0x400,  $-533

 Like C literal, but prefixed with ‘$’

 Encoded with 1, 2, 4, or 8 bytes 
depending on the instruction

 Register: 1 of 16 integer registers
 Examples:  %rax,  %r13

 But %rsp reserved for special use

 Others have special uses for particular 
instructions

 Memory: Consecutive bytes of memory 
at a computed address
 Simplest example:  (%rax)

 Various other “address modes”

%rax

%rcx

%rdx

%rbx

%rsi

%rdi

%rsp

%rbp

%rN

Moving Data

 General form:  mov_ source, destination
 Missing letter (_) specifies size of operands

 Note that due to backwards‐compatible support for 8086 
programs (16‐bit machines!), “word” means 16 bits = 2 bytes 
in x86 instruction names

 Lots of these in typical code

 movb src, dst
 Move 1‐byte “byte”

 movw src, dst
 Move 2‐byte “word”

 movl src, dst
 Move 4‐byte “long word”

 movq src, dst
 Move 8‐byte “quad word”

movq Operand Combinations

Source Dest Src, Dest C Analog

movq

Imm
Reg movq $0x4, %rax

Mem movq $-147, (%rax)

Reg
Reg movq %rax, %rdx

Mem movq %rax, (%rdx)

Mem Reg movq (%rax), %rdx

 Cannot do memory‐memory transfer with a single 
instruction

 How would you do it?

var_a = 0x4;

*p_a = -147;

var_d = var_a;

*p_d = var_a;

var_d = *p_a;

x86‐64 Introduction

 Arithmetic operations

 Memory addressing modes
 swap example

 Address computation instruction (lea)

Some Arithmetic Operations

 Binary (two‐operand) Instructions:



 Beware argument
order!

 No distinction
between signed
and unsigned
• Only arithmetic vs.
logical shifts

 How do you
implement 
“r3 = r1 + r2”?

Format Computation

addq src, dst dst = dst + src (dst += src)

subq src, dst dst = dst – src

imulq src, dst dst = dst * src signed mult

sarq src, dst dst = dst >> src Arithmetic

shrq src, dst dst = dst >> src Logical

shlq src, dst dst = dst << src (same as salq)

xorq src, dst dst = dst ^ src

andq src, dst dst = dst & src

orq src, dst dst = dst | src

Maximum of one 
memory operand

operand size specifier

Some Arithmetic Operations

 Unary (one‐operand) Instructions:

 See CSPP Section 3.5.5 for more instructions:  
mulq,  cqto,  idivq,  divq

Format Computation

incq dst dst = dst + 1 increment

decq dst dst = dst – 1 decrement

negq dst dst = –dst negate

notq dst dst = ~dst bitwise complement

Arithmetic Example

long simple_arith(long x, long y)
{

long t1 = x + y;
long t2 = t1 * 3;
return t2;

}

Register Use(s)

%rdi 1st argument (x)

%rsi 2nd argument (y)

%rax return value

y += x;
y *= 3;
long r = y; 
return r;

simple_arith:
addq %rdi, %rsi
imulq $3, %rsi
movq %rsi, %rax
ret

Example of Basic Addressing Modes

void swap(long *xp, long *yp) 
{

long t0 = *xp;
long t1 = *yp;
*xp = t1;
*yp = t0;

}

swap:
movq (%rdi), %rax
movq (%rsi), %rdx
movq %rdx, (%rdi)
movq %rax, (%rsi)
ret

Understanding  swap()

%rdi

%rsi

%rax

%rdx

Registers Memory

Register Variable

%rdi ⇔ xp

%rsi ⇔ yp

%rax ⇔ t0

%rdx ⇔ t1

void swap(long *xp, long *yp) 
{

long t0 = *xp;
long t1 = *yp;
*xp = t1;
*yp = t0;

}

swap:
movq (%rdi), %rax
movq (%rsi), %rdx
movq %rdx, (%rdi)
movq %rax, (%rsi)
ret

Understanding  swap()

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

%rdi

%rsi

%rax

%rdx

0x120

0x100

Registers Memory

123

456

123

swap:
movq (%rdi), %rax # t0 = *xp
movq (%rsi), %rdx # t1 = *yp
movq %rdx, (%rdi) # *xp =  t1
movq %rax, (%rsi) # *yp =  t0
ret

Understanding  swap()

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

%rdi

%rsi

%rax

%rdx

0x120

0x100

123

Registers Memory

123

456

123

swap:
movq (%rdi), %rax # t0 = *xp
movq (%rsi), %rdx # t1 = *yp
movq %rdx, (%rdi) # *xp =  t1
movq %rax, (%rsi) # *yp =  t0
ret

Understanding  swap()

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

%rdi

%rsi

%rax

%rdx

0x120

0x100

123

456

Registers Memory

123

456

123

swap:
movq (%rdi), %rax # t0 = *xp
movq (%rsi), %rdx # t1 = *yp
movq %rdx, (%rdi) # *xp =  t1
movq %rax, (%rsi) # *yp =  t0
ret

Understanding  swap()

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

%rdi

%rsi

%rax

%rdx

0x120

0x100

123

456

Registers Memory

123

456

456

swap:
movq (%rdi), %rax # t0 = *xp
movq (%rsi), %rdx # t1 = *yp
movq %rdx, (%rdi) # *xp =  t1
movq %rax, (%rsi) # *yp =  t0
ret

Understanding  swap()

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

%rdi

%rsi

%rax

%rdx

0x120

0x100

123

456

Registers Memory

123

123

456

swap:
movq (%rdi), %rax # t0 = *xp
movq (%rsi), %rdx # t1 = *yp
movq %rdx, (%rdi) # *xp =  t1
movq %rax, (%rsi) # *yp =  t0
ret

Memory Addressing Modes:  Basic

 Indirect: (R) Mem[Reg[R]]
 Data in register R specifies the memory address

 Like pointer dereference in C

 Example: movq (%rcx), %rax

 Displacement: D(R) Mem[Reg[R]+D]
 Data in register R specifies the start of some memory region

 Constant displacement D specifies the offset from that 
address

 Example: movq 8(%rbp), %rdx

Complete Memory Addressing Modes

 General:
 D(Rb,Ri,S) Mem[Reg[Rb]+Reg[Ri]*S+D]

• Rb: Base register (any register)

• Ri: Index register (any register except %rsp)

• S: Scale factor (1, 2, 4, 8) – why these numbers?

• D: Constant displacement value (a.k.a. immediate)

 Special cases  (see CSPP Figure 3.3 on p.181)
 D(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]+D] (S=1)

 (Rb,Ri,S) Mem[Reg[Rb]+Reg[Ri]*S] (D=0)

 (Rb,Ri) Mem[Reg[Rb]+Reg[Ri]] (S=1,D=0)

 (,Ri,S) Mem[Reg[Ri]*S] (Rb=0,D=0)
18

Address Computation Examples

%rdx

%rcx

0xf000

0x0100

Expression Address Computation Address

0x8(%rdx)

(%rdx,%rcx)

(%rdx,%rcx,4)

0x80(,%rdx,2)

D(Rb,Ri,S) →
Mem[Reg[Rb]+Reg[Ri]*S+D]

Address Computation Instruction

 leaq src, dst
 “lea” stands for load effective address

 src is address expression (any of the formats we’ve seen)

 dst is a register

 Sets dst to the address computed by the src expression 
(does not go to memory! – it just does math)

 Example:  leaq (%rdx,%rcx,4), %rax

 Uses:

 Computing addresses without a memory reference
• e.g. translation of  p = &x[i];

 Computing arithmetic expressions of the form  x+k*i+d
• Though k can only be 1, 2, 4, or 8

Example:  lea vs.  mov

21

0x120 

0x118

0x110 

0x108 

0x100 

Word
Address

Memory

123

0x10

0x1

0x400

0xF

0x8

Registers

%rax

%rbx

%rcx

%rdx

0x4

0x100

%rdi

%rsi

leaq (%rdx,%rcx,4), %rax
movq (%rdx,%rcx,4), %rbx
leaq (%rdx), %rdi
movq (%rdx), %rsi

Arithmetic Example

 Interesting Instructions
 leaq: “address” 

computation

 salq: shift

 imulq: multiplication

• Only used once!

long arith(long x, long y, long z)
{
long t1 = x + y;
long t2 = z + t1;
long t3 = x + 4;
long t4 = y * 48;
long t5 = t3 + t4;
long rval = t2 * t5;
return rval;

}

Register Use(s)

%rdi 1st argument (x)

%rsi 2nd argument (y)

%rdx 3rd argument (z)

arith:
leaq (%rdi,%rsi), %rax
addq %rdx, %rax
leaq (%rsi,%rsi,2), %rdx
salq $4, %rdx
leaq 4(%rdi,%rdx), %rcx
imulq %rcx, %rax
ret

Arithmetic Example

long arith(long x, long y, long z)
{
long t1 = x + y;
long t2 = z + t1;
long t3 = x + 4;
long t4 = y * 48;
long t5 = t3 + t4;
long rval = t2 * t5;
return rval;

}

Register Use(s)

%rdi x

%rsi y

%rdx z, t4

%rax t1, t2, rval

%rcx t5

arith:
leaq (%rdi,%rsi), %rax # rax/t1   = x + y
addq %rdx, %rax # rax/t2   = t1 + z
leaq (%rsi,%rsi,2), %rdx # rdx = 3 * y
salq $4, %rdx # rdx/t4   = (3*y) * 16
leaq 4(%rdi,%rdx), %rcx # rcx/t5   = x + t4 + 4
imulq %rcx, %rax # rax/rval = t5 * t2
ret

Peer Instruction Question

 Which of the following x86‐64 instructions correctly 
calculates %rax=9*%rdi?
 Vote at http://PollEv.com/justinh

A. leaq (,%rdi,9), %rax

B. movq (,%rdi,9), %rax

C. leaq (%rdi,%rdi,8), %rax

D. movq (%rdi,%rdi,8), %rax

E. We’re lost…

Summary

 Memory Addressing Modes: The addresses used for 
accessing memory in mov (and other) instructions can 
be computed in several different ways

 Base register, index register, scale factor, and displacement
map well to pointer arithmetic operations

 lea is address calculation instruction

 Does NOT actually go to memory

 Used to compute addresses or some arithmetic expressions

